// File: @openzeppelin/contracts/token/ERC20/ERC20.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts v4.4.0 (token/ERC20/ERC20.sol)

pragma solidity ^0.8.0;

import "./IERC20.sol";
import "./extensions/IERC20Metadata.sol";
import "../../utils/Context.sol";

/**
 * @dev Implementation of the {IERC20} interface.
 *
 * This implementation is agnostic to the way tokens are created. This means
 * that a supply mechanism has to be added in a derived contract using {_mint}.
 * For a generic mechanism see {ERC20PresetMinterPauser}.
 *
 * TIP: For a detailed writeup see our guide
 * https://forum.zeppelin.solutions/t/how-to-implement-erc20-supply-mechanisms/226[How
 * to implement supply mechanisms].
 *
 * We have followed general OpenZeppelin Contracts guidelines: functions revert
 * instead returning `false` on failure. This behavior is nonetheless
 * conventional and does not conflict with the expectations of ERC20
 * applications.
 *
 * Additionally, an {Approval} event is emitted on calls to {transferFrom}.
 * This allows applications to reconstruct the allowance for all accounts just
 * by listening to said events. Other implementations of the EIP may not emit
 * these events, as it isn't required by the specification.
 *
 * Finally, the non-standard {decreaseAllowance} and {increaseAllowance}
 * functions have been added to mitigate the well-known issues around setting
 * allowances. See {IERC20-approve}.
 */
contract ERC20 is Context, IERC20, IERC20Metadata {
    mapping(address => uint256) private _balances;

    mapping(address => mapping(address => uint256)) private _allowances;

    uint256 private _totalSupply;

    string private _name;
    string private _symbol;

    /**
     * @dev Sets the values for {name} and {symbol}.
     *
     * The default value of {decimals} is 18. To select a different value for
     * {decimals} you should overload it.
     *
     * All two of these values are immutable: they can only be set once during
     * construction.
     */
    constructor(string memory name_, string memory symbol_) {
        _name = name_;
        _symbol = symbol_;
    }

    /**
     * @dev Returns the name of the token.
     */
    function name() public view virtual override returns (string memory) {
        return _name;
    }

    /**
     * @dev Returns the symbol of the token, usually a shorter version of the
     * name.
     */
    function symbol() public view virtual override returns (string memory) {
        return _symbol;
    }

    /**
     * @dev Returns the number of decimals used to get its user representation.
     * For example, if `decimals` equals `2`, a balance of `505` tokens should
     * be displayed to a user as `5.05` (`505 / 10 ** 2`).
     *
     * Tokens usually opt for a value of 18, imitating the relationship between
     * Ether and Wei. This is the value {ERC20} uses, unless this function is
     * overridden;
     *
     * NOTE: This information is only used for _display_ purposes: it in
     * no way affects any of the arithmetic of the contract, including
     * {IERC20-balanceOf} and {IERC20-transfer}.
     */
    function decimals() public view virtual override returns (uint8) {
        return 18;
    }

    /**
     * @dev See {IERC20-totalSupply}.
     */
    function totalSupply() public view virtual override returns (uint256) {
        return _totalSupply;
    }

    /**
     * @dev See {IERC20-balanceOf}.
     */
    function balanceOf(address account) public view virtual override returns (uint256) {
        return _balances[account];
    }

    /**
     * @dev See {IERC20-transfer}.
     *
     * Requirements:
     *
     * - `recipient` cannot be the zero address.
     * - the caller must have a balance of at least `amount`.
     */
    function transfer(address recipient, uint256 amount) public virtual override returns (bool) {
        _transfer(_msgSender(), recipient, amount);
        return true;
    }

    /**
     * @dev See {IERC20-allowance}.
     */
    function allowance(address owner, address spender) public view virtual override returns (uint256) {
        return _allowances[owner][spender];
    }

    /**
     * @dev See {IERC20-approve}.
     *
     * Requirements:
     *
     * - `spender` cannot be the zero address.
     */
    function approve(address spender, uint256 amount) public virtual override returns (bool) {
        _approve(_msgSender(), spender, amount);
        return true;
    }

    /**
     * @dev See {IERC20-transferFrom}.
     *
     * Emits an {Approval} event indicating the updated allowance. This is not
     * required by the EIP. See the note at the beginning of {ERC20}.
     *
     * Requirements:
     *
     * - `sender` and `recipient` cannot be the zero address.
     * - `sender` must have a balance of at least `amount`.
     * - the caller must have allowance for ``sender``'s tokens of at least
     * `amount`.
     */
    function transferFrom(
        address sender,
        address recipient,
        uint256 amount
    ) public virtual override returns (bool) {
        _transfer(sender, recipient, amount);

        uint256 currentAllowance = _allowances[sender][_msgSender()];
        require(currentAllowance >= amount, "ERC20: transfer amount exceeds allowance");
        unchecked {
            _approve(sender, _msgSender(), currentAllowance - amount);
        }

        return true;
    }

    /**
     * @dev Atomically increases the allowance granted to `spender` by the caller.
     *
     * This is an alternative to {approve} that can be used as a mitigation for
     * problems described in {IERC20-approve}.
     *
     * Emits an {Approval} event indicating the updated allowance.
     *
     * Requirements:
     *
     * - `spender` cannot be the zero address.
     */
    function increaseAllowance(address spender, uint256 addedValue) public virtual returns (bool) {
        _approve(_msgSender(), spender, _allowances[_msgSender()][spender] + addedValue);
        return true;
    }

    /**
     * @dev Atomically decreases the allowance granted to `spender` by the caller.
     *
     * This is an alternative to {approve} that can be used as a mitigation for
     * problems described in {IERC20-approve}.
     *
     * Emits an {Approval} event indicating the updated allowance.
     *
     * Requirements:
     *
     * - `spender` cannot be the zero address.
     * - `spender` must have allowance for the caller of at least
     * `subtractedValue`.
     */
    function decreaseAllowance(address spender, uint256 subtractedValue) public virtual returns (bool) {
        uint256 currentAllowance = _allowances[_msgSender()][spender];
        require(currentAllowance >= subtractedValue, "ERC20: decreased allowance below zero");
        unchecked {
            _approve(_msgSender(), spender, currentAllowance - subtractedValue);
        }

        return true;
    }

    /**
     * @dev Moves `amount` of tokens from `sender` to `recipient`.
     *
     * This internal function is equivalent to {transfer}, and can be used to
     * e.g. implement automatic token fees, slashing mechanisms, etc.
     *
     * Emits a {Transfer} event.
     *
     * Requirements:
     *
     * - `sender` cannot be the zero address.
     * - `recipient` cannot be the zero address.
     * - `sender` must have a balance of at least `amount`.
     */
    function _transfer(
        address sender,
        address recipient,
        uint256 amount
    ) internal virtual {
        require(sender != address(0), "ERC20: transfer from the zero address");
        require(recipient != address(0), "ERC20: transfer to the zero address");

        _beforeTokenTransfer(sender, recipient, amount);

        uint256 senderBalance = _balances[sender];
        require(senderBalance >= amount, "ERC20: transfer amount exceeds balance");
        unchecked {
            _balances[sender] = senderBalance - amount;
        }
        _balances[recipient] += amount;

        emit Transfer(sender, recipient, amount);

        _afterTokenTransfer(sender, recipient, amount);
    }

    /** @dev Creates `amount` tokens and assigns them to `account`, increasing
     * the total supply.
     *
     * Emits a {Transfer} event with `from` set to the zero address.
     *
     * Requirements:
     *
     * - `account` cannot be the zero address.
     */
    function _mint(address account, uint256 amount) internal virtual {
        require(account != address(0), "ERC20: mint to the zero address");

        _beforeTokenTransfer(address(0), account, amount);

        _totalSupply += amount;
        _balances[account] += amount;
        emit Transfer(address(0), account, amount);

        _afterTokenTransfer(address(0), account, amount);
    }

    /**
     * @dev Destroys `amount` tokens from `account`, reducing the
     * total supply.
     *
     * Emits a {Transfer} event with `to` set to the zero address.
     *
     * Requirements:
     *
     * - `account` cannot be the zero address.
     * - `account` must have at least `amount` tokens.
     */
    function _burn(address account, uint256 amount) internal virtual {
        require(account != address(0), "ERC20: burn from the zero address");

        _beforeTokenTransfer(account, address(0), amount);

        uint256 accountBalance = _balances[account];
        require(accountBalance >= amount, "ERC20: burn amount exceeds balance");
        unchecked {
            _balances[account] = accountBalance - amount;
        }
        _totalSupply -= amount;

        emit Transfer(account, address(0), amount);

        _afterTokenTransfer(account, address(0), amount);
    }

    /**
     * @dev Sets `amount` as the allowance of `spender` over the `owner` s tokens.
     *
     * This internal function is equivalent to `approve`, and can be used to
     * e.g. set automatic allowances for certain subsystems, etc.
     *
     * Emits an {Approval} event.
     *
     * Requirements:
     *
     * - `owner` cannot be the zero address.
     * - `spender` cannot be the zero address.
     */
    function _approve(
        address owner,
        address spender,
        uint256 amount
    ) internal virtual {
        require(owner != address(0), "ERC20: approve from the zero address");
        require(spender != address(0), "ERC20: approve to the zero address");

        _allowances[owner][spender] = amount;
        emit Approval(owner, spender, amount);
    }

    /**
     * @dev Hook that is called before any transfer of tokens. This includes
     * minting and burning.
     *
     * Calling conditions:
     *
     * - when `from` and `to` are both non-zero, `amount` of ``from``'s tokens
     * will be transferred to `to`.
     * - when `from` is zero, `amount` tokens will be minted for `to`.
     * - when `to` is zero, `amount` of ``from``'s tokens will be burned.
     * - `from` and `to` are never both zero.
     *
     * To learn more about hooks, head to xref:ROOT:extending-contracts.adoc#using-hooks[Using Hooks].
     */
    function _beforeTokenTransfer(
        address from,
        address to,
        uint256 amount
    ) internal virtual {}

    /**
     * @dev Hook that is called after any transfer of tokens. This includes
     * minting and burning.
     *
     * Calling conditions:
     *
     * - when `from` and `to` are both non-zero, `amount` of ``from``'s tokens
     * has been transferred to `to`.
     * - when `from` is zero, `amount` tokens have been minted for `to`.
     * - when `to` is zero, `amount` of ``from``'s tokens have been burned.
     * - `from` and `to` are never both zero.
     *
     * To learn more about hooks, head to xref:ROOT:extending-contracts.adoc#using-hooks[Using Hooks].
     */
    function _afterTokenTransfer(
        address from,
        address to,
        uint256 amount
    ) internal virtual {}
}

// File: @openzeppelin/contracts/token/ERC20/extensions/IERC20Metadata.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts v4.4.0 (token/ERC20/extensions/IERC20Metadata.sol)

pragma solidity ^0.8.0;

import "../IERC20.sol";

/**
 * @dev Interface for the optional metadata functions from the ERC20 standard.
 *
 * _Available since v4.1._
 */
interface IERC20Metadata is IERC20 {
    /**
     * @dev Returns the name of the token.
     */
    function name() external view returns (string memory);

    /**
     * @dev Returns the symbol of the token.
     */
    function symbol() external view returns (string memory);

    /**
     * @dev Returns the decimals places of the token.
     */
    function decimals() external view returns (uint8);
}

// File: @openzeppelin/contracts/token/ERC20/IERC20.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts v4.4.0 (token/ERC20/IERC20.sol)

pragma solidity ^0.8.0;

/**
 * @dev Interface of the ERC20 standard as defined in the EIP.
 */
interface IERC20 {
    /**
     * @dev Returns the amount of tokens in existence.
     */
    function totalSupply() external view returns (uint256);

    /**
     * @dev Returns the amount of tokens owned by `account`.
     */
    function balanceOf(address account) external view returns (uint256);

    /**
     * @dev Moves `amount` tokens from the caller's account to `recipient`.
     *
     * Returns a boolean value indicating whether the operation succeeded.
     *
     * Emits a {Transfer} event.
     */
    function transfer(address recipient, uint256 amount) external returns (bool);

    /**
     * @dev Returns the remaining number of tokens that `spender` will be
     * allowed to spend on behalf of `owner` through {transferFrom}. This is
     * zero by default.
     *
     * This value changes when {approve} or {transferFrom} are called.
     */
    function allowance(address owner, address spender) external view returns (uint256);

    /**
     * @dev Sets `amount` as the allowance of `spender` over the caller's tokens.
     *
     * Returns a boolean value indicating whether the operation succeeded.
     *
     * IMPORTANT: Beware that changing an allowance with this method brings the risk
     * that someone may use both the old and the new allowance by unfortunate
     * transaction ordering. One possible solution to mitigate this race
     * condition is to first reduce the spender's allowance to 0 and set the
     * desired value afterwards:
     * https://github.com/ethereum/EIPs/issues/20#issuecomment-263524729
     *
     * Emits an {Approval} event.
     */
    function approve(address spender, uint256 amount) external returns (bool);

    /**
     * @dev Moves `amount` tokens from `sender` to `recipient` using the
     * allowance mechanism. `amount` is then deducted from the caller's
     * allowance.
     *
     * Returns a boolean value indicating whether the operation succeeded.
     *
     * Emits a {Transfer} event.
     */
    function transferFrom(
        address sender,
        address recipient,
        uint256 amount
    ) external returns (bool);

    /**
     * @dev Emitted when `value` tokens are moved from one account (`from`) to
     * another (`to`).
     *
     * Note that `value` may be zero.
     */
    event Transfer(address indexed from, address indexed to, uint256 value);

    /**
     * @dev Emitted when the allowance of a `spender` for an `owner` is set by
     * a call to {approve}. `value` is the new allowance.
     */
    event Approval(address indexed owner, address indexed spender, uint256 value);
}

// File: @openzeppelin/contracts/utils/Context.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts v4.4.0 (utils/Context.sol)

pragma solidity ^0.8.0;

/**
 * @dev Provides information about the current execution context, including the
 * sender of the transaction and its data. While these are generally available
 * via msg.sender and msg.data, they should not be accessed in such a direct
 * manner, since when dealing with meta-transactions the account sending and
 * paying for execution may not be the actual sender (as far as an application
 * is concerned).
 *
 * This contract is only required for intermediate, library-like contracts.
 */
abstract contract Context {
    function _msgSender() internal view virtual returns (address) {
        return msg.sender;
    }

    function _msgData() internal view virtual returns (bytes calldata) {
        return msg.data;
    }
}

// File: contracts/CASINOTokenLauncherMainnet.sol



// SPDX-License-Identifier: GPL-3.0-only
pragma solidity 0.8.9;

import { IERC20 } from "@openzeppelin/contracts/token/ERC20/IERC20.sol";

import { FixedSupplyToken } from "./FixedSupplyToken.sol";
import { IUniswapV2Router } from "./IUniswapV2Router.sol";

contract CASINOTokenLauncherMainnet
{
	struct Recipient {
		address to;
		uint256 value;
		uint256 amount;
	}

	address constant OWNER = 0x2F80922CF7350e06F4924766Cb7EEEC783c1C8ce;

	address constant RECEIVER = 0xB0632a01ee778E09625BcE2a257e221b49E79696;

	address constant ROUTER = 0x7a250d5630B4cF539739dF2C5dAcb4c659F2488D; // UniswapV2

	address constant WETH = 0xC02aaA39b223FE8D0A0e5C4F27eAD9083C756Cc2; // WETH

	address constant FURNACE = 0x000000000000000000000000000000000000dEaD;

	string constant NAME = "CASINO";

	uint256 constant SUPPLY = 10_000_000_000e18; // 10B tokens

	uint256 constant DEDUCTION = SUPPLY * 4.5e16 / 100e16; // 4.5% of supply

	uint256 constant LIQUIDITY = 2.5e18; // 2.5 ETH

	uint256 constant PURCHASE = 1.5e18; // 1.5 ETH

	address public token;

	Recipient[] public airdrop;

	constructor()
	{
		airdrop.push(Recipient({ to: 0x985C56eB1e8951329aa862570AEB584B074bEB10, value:   0.5e18, amount: 0 }));
		airdrop.push(Recipient({ to: 0x80F2dCC36D9548F97A14a3bF73D992FB614e45f4, value:   0.5e18, amount: 0 }));
		airdrop.push(Recipient({ to: 0x2165fa4a32B9c228cD55713f77d2e977297D03e8, value:   0.5e18, amount: 0 }));
		airdrop.push(Recipient({ to: 0xc1a17F2fd7DB9adf67De8a8F331d682923090Feb, value:   0.6e18, amount: 0 }));
		airdrop.push(Recipient({ to: 0x6A645f11716Fe2F0b308515FFbE383b490a4bFCE, value:     1e18, amount: 0 }));
		airdrop.push(Recipient({ to: 0xEE79a26aE73e26224B49eC6fE3A62B52424B9D0b, value:     1e18, amount: 0 }));
		airdrop.push(Recipient({ to: 0xA367B1895deE09AEE1E1FdbD1027e283c1198046, value:   0.5e18, amount: 0 }));
		airdrop.push(Recipient({ to: 0x99335137cB4621498ef0719bc8b871003e7bdc44, value:  0.14e18, amount: 0 }));
		airdrop.push(Recipient({ to: 0x94F8499D9F0aa65b41Cf2f8ED09A2260530ED269, value:     1e18, amount: 0 }));
		airdrop.push(Recipient({ to: 0x906B2402b7b569e68A04B11fb7d5A576BC11d51D, value:   0.7e18, amount: 0 }));
		airdrop.push(Recipient({ to: 0xcD8dDeE99C0c4Be4cD699661AE9c00C69D1Eb4A8, value:   0.5e18, amount: 0 }));
		airdrop.push(Recipient({ to: 0x1D102Cc151C2CD40C9ceDa6394b7649cDDD260B4, value: 0.017e18, amount: 0 }));
		airdrop.push(Recipient({ to: 0x93e5cEda644Da2b97c6b8B014756E1cA0927aDa7, value:   0.3e18, amount: 0 }));
		airdrop.push(Recipient({ to: 0xBe7E848Db82C1bE21fDD6AaCdFC51c900B2682d6, value:   0.4e18, amount: 0 }));
		airdrop.push(Recipient({ to: 0x76B5cbdB978122e147F5105cF2801F99C87c0F89, value:     1e18, amount: 0 }));
		uint256 _value = 0;
		for (uint256 _i = 0; _i < airdrop.length; _i++) {
			Recipient storage _recipient = airdrop[_i];
			_value += _recipient.value;
		}
		/*
		airdrop0 = 7.657
		reserve0 = 2.5
		amount0 = 1.5
		reserve0' = reserve0 + amount0

		reserve1 = 9.55B - airdrop1
		amount1 = (amount0 * 997 * reserve1) / (reserve0 * 1000 + amount0 * 997)
		reserve1' = reserve1 - amount1

		airdrop1 = airdrop0 * reserve1' / reserve0'

		Deduction:

		reserve1' = reserve1 - amount1
		reserve1' = reserve1 - (amount0 * 997 * reserve1) / (reserve0 * 1000 + amount0 * 997)
		reserve1' = ((reserve0 * 1000 + amount0 * 997) * reserve1 - (amount0 * 997 * reserve1)) / (reserve0 * 1000 + amount0 * 997)
		reserve1' = ((reserve0 * 1000 * reserve1 + amount0 * 997 * reserve1) - (amount0 * 997 * reserve1)) / (reserve0 * 1000 + amount0 * 997)
		reserve1' = (reserve0 * 1000 * reserve1) / (reserve0 * 1000 + amount0 * 997)
		reserve1' = (reserve0 * 1000 * (9.55B - airdrop1)) / (reserve0 * 1000 + amount0 * 997)
		reserve1' = (reserve0 * 1000 * 9.55B - reserve0 * 1000 * airdrop1) / (reserve0 * 1000 + amount0 * 997)
		(reserve0 * 1000 + amount0 * 997) * reserve1' = reserve0 * 1000 * 9.55B - reserve0 * 1000 * airdrop1
		(reserve0 * 1000 + amount0 * 997) * reserve1' = reserve0 * 1000 * 9.55B - reserve0 * 1000 * (airdrop0 * reserve1' / reserve0')
		(reserve0 * 1000 + amount0 * 997) * reserve1' * reserve0' = reserve0 * 1000 * 9.55B * reserve0' - reserve0 * 1000 * airdrop0 * reserve1'
		(reserve0 * 1000 + amount0 * 997) * reserve1' * reserve0' + reserve0 * 1000 * airdrop0 * reserve1' = reserve0 * 1000 * 9.55B * reserve0'
		(((reserve0 * 1000 + amount0 * 997) * reserve0') + (reserve0 * 1000 * airdrop0)) * reserve1' = reserve0 * 1000 * 9.55B * reserve0'
		((reserve0 * 1000 * reserve0' + amount0 * 997 * reserve0') + (reserve0 * 1000 * airdrop0)) * reserve1' = reserve0 * 1000 * 9.55B * reserve0'
		(reserve0 * 1000 * reserve0' + amount0 * 997 * reserve0' + reserve0 * 1000 * airdrop0) * reserve1' = reserve0 * 1000 * 9.55B * reserve0'
		reserve1' = (reserve0 * 1000 * 9.55B * reserve0') / (reserve0 * 1000 * reserve0' + amount0 * 997 * reserve0' + reserve0 * 1000 * airdrop0)
		reserve1' = (reserve0 * 1000 * 9.55B * reserve0') / (amount0 * 997 * reserve0' + reserve0 * 1000 * (reserve0' + airdrop0))
		reserve1' = (2.5 * 1000 * 9.55B * 4) / (1.5 * 997 * 4 + 2.5 * 1000 * (4 + airdrop0))
		*/
		uint256 _reserveValue = LIQUIDITY + PURCHASE;
		uint256 _reserveAmount = (LIQUIDITY * 1000 * (SUPPLY - DEDUCTION) * _reserveValue) / (PURCHASE * 997 * _reserveValue + LIQUIDITY * 1000 * (_reserveValue + _value));
		for (uint256 _i = 0; _i < airdrop.length; _i++) {
			Recipient storage _recipient = airdrop[_i];
			_recipient.amount = (_recipient.value * _reserveAmount * 90) / (_reserveValue * 100);
		}
	}

	function launch() external payable returns (address _token)
	{
		require(token == address(0), "already launched");
		require(msg.sender == OWNER || msg.sender == RECEIVER, "access denied");
		require(msg.value == LIQUIDITY + PURCHASE, "invalid value");
		token = address(new FixedSupplyToken(NAME, NAME, SUPPLY));
		IERC20(token).transfer(RECEIVER, DEDUCTION);
		uint256 _liquidity = SUPPLY - DEDUCTION;
		for (uint256 _i = 0; _i < airdrop.length; _i++) {
			Recipient storage _recipient = airdrop[_i];
			IERC20(token).transfer(_recipient.to, _recipient.amount);
			_liquidity -= _recipient.amount;
		}
		IERC20(token).approve(ROUTER, _liquidity);
		IUniswapV2Router(ROUTER).addLiquidityETH{value: LIQUIDITY}(token, _liquidity, _liquidity, LIQUIDITY, RECEIVER, block.timestamp);
		address[] memory _path = new address[](2);
		_path[0] = WETH;
		_path[1] = token;
		IUniswapV2Router(ROUTER).swapExactETHForTokens{value: PURCHASE}(0, _path, FURNACE, block.timestamp);
		return token;
	}
}

// File: contracts/FixedSupplyToken.sol



// SPDX-License-Identifier: GPL-3.0-only
pragma solidity 0.8.9;

import { ERC20 } from "@openzeppelin/contracts/token/ERC20/ERC20.sol";

contract FixedSupplyToken is ERC20
{
	constructor(string memory _name, string memory _symbol, uint256 _supply)
		ERC20(_name, _symbol)
	{
		_mint(msg.sender, _supply);
	}
}

// File: contracts/IUniswapV2Router.sol



// SPDX-License-Identifier: GPL-3.0-only
pragma solidity 0.8.9;

interface IUniswapV2Router
{
	function factory() external view returns (address _factory);

	function addLiquidityETH(address _token, uint256 _amountTokenDesired, uint256 _amountTokenMin, uint256 _amountETHMin, address _to, uint256 _deadline) external payable returns (uint256 _amountToken, uint256 _amountETH, uint256 _liquidity);
	function swapExactETHForTokens(uint256 _amountOutMin, address[] calldata _path, address _to, uint256 _deadline) external payable returns (uint256[] memory _amounts);
	function swapExactTokensForTokensSupportingFeeOnTransferTokens(uint256 _amountIn, uint256 _amountOutMin, address[] calldata _path, address _to, uint256 _deadline) external;
}